$1.00

DIGITAL EQUIPMENT CORPORATION

education

BASIC Matrix Operations



BASIC MATRIX OPERATIONS

The curriculum material and programs in this booklet introduce the idea of a matrix. They go on to discuss matrix operations of addition, subtraction, multiplication by a scaler, and matrix multiplication. The last section covers several contemporary applications of matrix manipulations.

Many versions of BASIC have special "matrix" commands that simplfy working with matrices. This booklet does not use these commands. There are two reasons for this. First, professional programmers do not use them, nor are they available in computer languages other than BASIC. Second, you will obtain much more insight into the mathematics of matrices by writing programs that manipulate them in the same way that the definitions say you should.

CONTENTS

This booklet was prepared by and is reprinted with the permission of Project SOLO, Department of Computer Science, University of Pittsburgh, Pittsburgh, PA 15213.

DEC EDUCATIONAL PUBLICATIONS

Other publications in the continuing series of curriculum material published by DEC for use with EduSystems 10 through 90 are listed below. Please inquire directly for prices on classroom quantities.

Additional publications may be obtained from:

Program Library
Digital Equipment Corporation
Maynard, Massachusetts 01754

Populution: Self-teaching BASIC Workbook	$2.00
BASIC Application Programs	
Set 1 (15 Programs)	1.00
Set 2 (19 Programs)	1.25
Set 3 (12 Programs)	.75
Set 4 (Plotter Programs)	1.00
BASIC Matrix Operations	1.00
BASIC Simulation Programs	
Vol I (Biology and Earth Science Programs) } Vol II (Chemistry Programs)	4.00
Vol III (Mathematics Programs) } Vol IV (Physics Programs	4.00
Vol V (Social Studies Programs) } Vol VI (Teacher Assistance)	3.00
Problems for Computer Mathematics (80 descriptive problems for all levels)	1.25

Elementary Matrix Operations

PROJECT SOLO
Department of Computer Science
University of Pittsburgh
(15213)

Module #0107



● This module introduces the idea of a <u>matrix</u>, and the elementary matrix operations of addition (page 5) and subtraction (page 7), and multiplication by a scalar (page 9).

● Module # 0035 covers matrix multiplication. Module #0045 discusses several modern applications of matrix multiplication.

Elementary MATRIX Operations

The map on the cover of this module shows the streets and avenues of a section of the town of Munchin. As can be seen, there are employees of Col. Mayer's finger-licking Hot Dog Vending Company stationed at strategic intersections in the town. Being the business man that he is, Col. Mayer is interested in keeping an account of the number of sales of each of these vendors. He represents these sales figures in an ordered rectangular array as follows:

SALES FOR MONTH OF APRIL

	30th St.	32nd St.	34th St.	36th St.	38th St.
4th Avenue	0	877	0	1,996	0
5th Avenue	1,652	0	2,008	0	1,765
7th Avenue	3,077	2,437	0	2,981	0
12th Avenue	3,618	3,051	4,171	4,632	3,987

This rectangular array is called a matrix (pronounced MAY-TRICKS). Each row of this matrix represents locations along a certain avenue. Each column of the matrix represents locations along a certain street. The entries of the matrix, which are called elements, represent the number of hot dogs sold by each vendor during the month of April.

Notice that the elements of the matrix show not only the number of sales but also the location of the vendor. For instance the entry 877 is the number of hot dogs sold by the vendor located at 4th Avenue and 32nd Street.

****
QUICK QUIZ
****

1. What is the location of the vendor whose sales totaled 4,171 for the month of April?

ANSWER: _______ Avenue and _______ Street.

2. How many hot dogs were sold in April by the vendor at 12th Avenue and 36th Street?

ANSWER: ______ hot dogs.

The 0 entries mean no sales because either there is no vendor at that corner or the vendor sold 0 hot dogs.

In the above matrix there are 4 <u>rows</u> (one for each of the avenues) and 5 <u>columns</u> (one for each street). The <u>size</u> of a matrix is given by the number of rows and columns. In this case the matrix is said to have size 4 by 5. In general, the size of a matrix is given as the <u>number of rows</u> by the <u>number of columns</u>.

MATRIX ADDITION

Another month has passed and Colonel Mayer has collected the sales of hot dogs by his vendors for the month of May. Here are the numbers:

SALES FOR MONTH OF MAY

	30th St.	32nd St.	34th St.	36th St.	38th St.
4th Avenue	0	1,002	0	1,890	0
5th Avenue	1,723	0	2,971	0	1,887
7th Avenue	4,007	2,983	0	3,623	0
12th Avenue	3,773	3,076	5,380	5,179	4,837

Suppose we now want a table showing the <u>total</u> number of hot dogs sold by each vendor for the 2 months of April and May? It is probably obvious to you that the way to get this table is to add together each vendor's sales for both months. Doing this for the <u>whole</u> <u>chart</u> is called the operation of <u>matrix</u> <u>addition</u>.

Before going ahead and doing this, we first will introduce a more compact notation for matrices,* used by mathematicians.. We will discard everything except the elements of the matrix, enclose them in square brackets, and give them a name as follows (capital letters are usually used):

$$A=\begin{bmatrix} 0 & 877 & 0 & 1{,}996 & 0 \\ 1{,}652 & 0 & 2{,}008 & 0 & 1{,}765 \\ 3{,}077 & 2{,}437 & 0 & 2{,}981 & 0 \\ 3{,}618 & 3{,}051 & 4{,}171 & 4{,}632 & 3{,}987 \end{bmatrix}$$

This is the matrix A which describes the sales for April

$$M=\begin{bmatrix} 0 & 1{,}002 & 0 & 1{,}890 & 0 \\ 1{,}723 & 0 & 2{,}971 & 0 & 1{,}887 \\ 4{,}007 & 2{,}983 & 0 & 3{,}623 & 0 \\ 3{,}773 & 3{,}076 & 5{,}380 & 5{,}179 & 4{,}837 \end{bmatrix}$$

This is the matrix M which describes the sales for May

To answer the last question (what are total sales of each vendor for April and May) we simply add the matrices A and M. This means that each element in A is added to the <u>corresponding</u> element in M.

---

*Matrices is the plural of matrix; it is pronounced MAY-TRA-SEES.

The matrix A:

0	877	0	1,996	0
1,652	0	2,008	0	1,765
3,077	2,437	0	2,981	0
3,618	3,051	4,171	4,632	3,987

\+ The matrix M:

0	1,002	0	1,890	0
1,723	0	2,971	0	1,887
4,007	2,983	0	3,623	0
3,773	3,076	5,380	5,179	4,837

The matrix A+M =

0+0	877+1,002	0+0	1,996+1,890	0+0
1,652+1,723	0+0	2,008+2,971	0+0	1,765+1,887
3,077+4,007	2,437+2,983	0+0	2,981+3,623	0+0
3,618+3,773	3,051+3,076	4,171+5,380	4,632+5,179	3,987+4,837

=

0	1,879	0	3,886	0
3,375	0	4,979	0	3,652
______	______	______	______	______
______	______	______	______	______

****
EXERCISE
****

Fill in the missing entries in the sum matrix shown above. (Hard work? Cheer-up. You'll learn to use the computer for this purpose soon.)

MATRIX SUBTRACTION

Colonel Mayer wants to know the difference in hot dog sales between May and April for each vendor so that he can tell which location is showing the greatest boom (increase) in business.

The answer to this problem involves the matrix operation of subtraction. Matrix subtraction is similar to matrix addition, except that each element of A will be <u>subtracted</u> from the <u>corresponding</u> element of M. Thus,

$$M-A=\begin{bmatrix} 0 & 1,002 & 0 & 1,890 & 0 \\ 1,723 & 0 & 2,971 & 0 & 1,887 \\ 4,007 & 2,983 & 0 & 3,623 & 0 \\ 3,773 & 3,076 & 5,380 & 5,179 & 4,837 \end{bmatrix}$$

← The matrix M

$$-\begin{bmatrix} 0 & 877 & 0 & 1,996 & 0 \\ 1,652 & 0 & 2,008 & 0 & 1,765 \\ 3,077 & 2,437 & 0 & 2,981 & 0 \\ 3,618 & 2,051 & 4,171 & 4,632 & 3,087 \end{bmatrix}$$

← The matrix A

$$M-A=\begin{bmatrix} 0-0 & 1,002-\ 877 & 0-0 & 1,890-1,996 & 0-0 \\ 1,723-1,653 & 0-0 & 2,971-2,008 & 0-0 & 1,887-1,765 \\ 4,007-3,077 & 2,983-2,437 & 0-0 & 3,623-2,981 & 0-0 \\ 3,773-3,518 & 3,076-3,051 & 5,380-4,171 & 5,179-4,632 & 4,837-3,987 \end{bmatrix}$$

$$=\begin{bmatrix} 0 & 125 & 0 & -106 & 0 \\ 70 & 0 & 963 & 0 & 122 \\ 930 & \_\_\_ & 0 & \_\_\_ & 0 \\ 155 & \_\_\_ & \_\_\_ & \_\_\_ & 850 \end{bmatrix}$$

EXERCISE

1. Complete the above "difference" matrix.

In performing the matrix operations of addition and subtraction, notice that the 2 matrices being added or subtracted must be of the same size.

MATRIX MULTIPLICATION BY A SCALAR

With the approach of the summer months and the baseball season, Col. Mayer is predicting a 20% increase over the May sales for the month of June. What number of hot dogs does he predict each vendor to sell during the month of June?

The answer to this question involves the matrix operation of multiplication by a scalar. This means that each element in a matrix will be multiplied by the same number which is called a scalar. The scalar multiplier for our example is 1.20. We can say that each element in June is predicted to be 1.2 times as big as the corresponding element in May. The compact way of saying this with matrices is

J=1.2*M.

Which means:

J=

1.20*0	1.20*1,002	1.20*0	1.20* 1,890	1.20*0
1.20*1,723	1.20*0	______	______	______
1.20*4,007	______	1.20*0	______	1.20*0
1.20*2,773	1.20*3,076	1.20*5,380	1.20*5,179	1.20*4,837

J=

0	______	0	______	0
______	0	______	______	______
______	______	0	______	0
______	______	______	______	______

EXERCISE (What, again!)

Complete the matrix J=1.20*M shown on the previous page. (If you get tired, read on and find out how to use the computer to do this.)

Using the Computer to Add and Subtract Matrices and to Multiply Matrices by a Scalar

> **"IMPORTANT NOTE:"** Many versions of BASIC have special "matrix" commands that simplify working with matrices. However in this module we will not use these commands. There are two reasons for this. First, professional programmers do not use them, nor are they available in computer languages other than BASIC. Second, you will obtain much more insight into the mathematics of matrices by writing programs that manipulate them in the same way that the definitions say you should.

This module assumes that you understand what a subscripted variable is, and how such variables are used in BASIC.



STORING A MATRIX IN THE COMPUTER

Since an element of a matrix is located by its row number and column number (see page 2), the best way to store matrix elements is to use variables (locations) which have two subscripts. Then we can store the element which is in row i and column j in the location called A(I,J). For example, A(1,2) will be the variable name we give to the location where we store the matrix element in the first row, second column of A.

INPUT OF MATRIX ELEMENTS

The program segment shown below allows you to input the elements of a 2 by 3 (2×3) matrix. It expects you to input the numbers left to right, row by row.

```
105 DIM A(2,3)
110 FOR I=1 TO 2
115 FOR J=1 TO 3
120 INPUT A(I,J)
125 NEXT J
130 NEXT I
1000 END
RUN
?12
?-2
?14
?0
?33
?19
```

After you run this program, the memory of the computer looks like the following:

A(1,1) 12	A(1,2) -2	A(1,3) 14
A(2,1) 0	A(2,2) 33	A(2,3) 19

PRINTING OUT ELEMENTS

To print out the above matrix elements row by row we can add the following statements to the above program:

```
910 FOR I=1 TO 2
915 FOR J=1 TO 3
920 PRINT A(I,J);   <-- The semi-colon forces all the elements
                        controlled by the inner FOR Loop to be
                        printed on one line.
925 NEXT J
930 PRINT (cr)      <-- The extra PRINT statement forces the com-
                        puter to print the next row on a new line.
935 NEXT I
```

SAMPLE PROGRAM - Enter and run this program for the Hot Dog Program (on p. 4).

Matrix Addition: C=A+B

```
1 DIM A(4,5), B(4,5), C(4,5)
5 PR. "THIS PROGRAM ADDS TWO 4 by 5 MATRICES"
10 PR. "INPUT THE ELEMENTS OF MATRIX A"
15 FOR I=1 TO 4
20 FOR J=1 TO 5
25 INPUT A(I,J)
30 NEXT J
35 NEXT I
40 PR. "NOW INPUT THE ELEMENTS OF MATRIX B"
45 FOR I=1 TO 4
50 FOR J=1 TO 5
55 INPUT B(I,J)
60 NEXT J
65 NEXT I
70 FOR I=1 TO 4
75 FOR J=1 TO 5
80 C(I,J)= A(I,J)+B(I,J)
85 NEXT J
90 NEXT
95 PR. "THE SUM MATRIX IS:
100 FOR I=1 TO 4
105 FOR J=1 TO 5
110 PRINT C(I,J);
115 NEXT J
120 PRINT
125 NEXT I
130 END
```

Lines 10–35: Input Matrix A

Lines 40–65: Input Matrix B

Lines 70–90: This segment of the program adds each element of B to the corresponding element of A, and puts the sum in the cor-responding position of Matrix C

Lines 95–130: This prints out the Matrix C which contains the sum elements

?????????????????????

CHALLENGE PROGRAM #1

?????????????????????

Write a program to calculate C=A+B for matrices that can have any size up to and including 10 rows and 10 columns.

HINTS:

a. Dimension A, B, and C as 10 by 10

b. Allow the user to input the number of ROWS = M

c. Allow the user to input the number of COLUMNS = N

d. Change all the FOR loops to go TO M or TO N

??????????????????????
CHALLENGE PROGRAM #2
??????????????????????

Test the above program on your own 10 × 10 matrices.

??????????????????????
CHALLENGE PROGRAM #3
??????????????????????

Write a program which inputs a Matrix A and inputs a scalar factor J. The program should output the Matrix D which is the scalar product of J times M, i.e.



Test this program on the Hot Dog Problem on page 6.

??????????????????????
CHALLENGE PROGRAM #4
??????????????????????

(Confidential Note: We will throw in some technical language in this problem which will make it appear complicated. But the mathematics (and computing) is no harder than that used in the Hot Dog problem.)

The chart below shows the observed frequency values found in a series of pathologically verified cases of ulcer and cancer of the stomach:*

F=

	Achlorhydria	Hypochlorhydria	Normal	Hyperchlorhydria
Chronic Ulcer	3	7	35	9
Stomach Cancer	22	2	6	0

The expected frequency values are given in the table E below:

E=

	Achlorhydria	Hypochlorhydria	Normal	Hyperchlorhydria
Chronic Ulcer	16	5.8	26.4	5.8
Stomach Cancer	9	3.2	14.6	3.2

These tables can also be represented as matrices:

$$F=\begin{bmatrix} 3 & 7 & 35 & 9 \\ 22 & 2 & 6 & 0 \end{bmatrix}$$ (observed values)

$$E=\begin{bmatrix} 16 & 5.8 & 26.4 & 5.8 \\ 9 & 3.2 & 14.6 & 3.2 \end{bmatrix}$$ (expected values)

For statistical purposes we would like a table showing the differences between the observed and expected values. This means that you should modify your program to output the matrix F-E. HINT: As you might guess, all you have to do is modify the matrix addition program on page 9 so that line 80 uses a 'minus sign' (-). (Also change line 95.)

*See the book "Facts from Figures" by M. J. Horoney, pp. 246-270.

Partial Output for Program #4

$$F-E=\begin{bmatrix} -13.0 & 1.2 & \_\_\_\_ & \_\_\_\_ \\ \_\_\_\_ & \_\_\_\_ & \_\_\_\_ & \_\_\_\_ \end{bmatrix}$$

??????????????????????
CHALLENGE PROGRAM #5
??????????????????????

Modify the above program so that a matrix of relative differences between observed and expected values is printed out, followed by a matrix of percentage differences.

Example for one element: (let's take the first row, second column)

F(1,2) = 7 (observed)

E(1,2) = 5.8 (expected)

The relative difference by which the observed value differs from the expected value for these particular elements is calculated by dividing the difference between F and E by E:

(7-5.8)/5.8 = 0.20689 = RELATIVE DIFFERENCE

The percentage difference is obtained by multiplying the relative difference by 100:

0.20689*100 = 20.689% = PERCENTAGE DIFFERENCE

Now go ahead and do this for all elements in the matrices, using the computer as suggested on the next page.

HINT 1: Modify line 80 (p. 9) so that the Matrix C will contain relative differences.

C(I,J) = (F(I,J)-E(I,J))/E(I,J)

HINT 2: For percentage differences do a scalar multiplication by 100. You can use the Matrix C "over again" (after printing the relative difference) by letting C(I,J) = 100*C(I,J) within a FOR loop.

HINT:3: Since you will want to print out two matrices (relative differences and percentage differences), but since they both will be called C (see HINT 2 above), it would be a good idea to put the print statements in a sub-routine as follows:



Matrix Multiplication

PROBLEM

$$\overset{A}{\begin{bmatrix} 5 & 4 & 2 \\ 3 & 9 & 8 \\ 6 & 2 & 1 \end{bmatrix}} * \overset{B}{\begin{bmatrix} 2 & 7 \\ 5 & 3 \\ 6 & 4 \end{bmatrix}} = \overset{C}{\begin{bmatrix} ? & ? \\ ? & ? \\ ? & ? \end{bmatrix}}$$



SOLUTION



Step 1.

$$\overset{A}{\begin{bmatrix} 5 & 4 & 2 \\ 3 & 9 & 8 \\ 6 & 2 & 1 \end{bmatrix}} * \overset{B}{\begin{bmatrix} 2 & 7 \\ 5 & 3 \\ 6 & 4 \end{bmatrix}} = \overset{C}{\begin{bmatrix} 42 & ? \\ 99 & ? \\ 28 & ? \end{bmatrix}}$$

2 5 6

(5•2) + (4•5) + (2•6)

(3•2) + (9•5) + (8•6)

(6•2) + (2•5) + (1•6)

Step 2.

$$\overset{A}{\begin{bmatrix} 5 & 4 & 2 \\ 3 & 9 & 8 \\ 6 & 2 & 1 \end{bmatrix}} * \overset{B}{\begin{bmatrix} 2 & 7 \\ 5 & 3 \\ 6 & 4 \end{bmatrix}} = \overset{C}{\begin{bmatrix} 42 & 55 \\ 99 & 80 \\ 28 & 52 \end{bmatrix}}$$

7 3 4

(5•7) + (4•3) + (2•4)

(3•7) + (9•3) + (8•4)

(6•7) + (2•3) + (1•4)



PROJECT SOLO / Dept. of Computer Science / Univ. of Pittsburgh(15213)
Frank Wimberly(PS) / Module #0035

MATRIX MULTIPLICATION

INTRODUCTION

The need to develop a compact notation for manipulating systems of linear equations led 19th century mathematicians to the discovery and use of matrices. (*Deep question: Are mathematical concepts discovered or invented?*) For the definition of a matrix and a discussion of matrix addition and multiplication of matrices by scalars *see the module "ELEMENTRAY MATRIX OPERATIONS"*. The present module is a presentation of another important matrix operation--matrix multiplication.

Multiplication of matrices was first suggested by research in the theory of linear equations, but it has turned out to be an important concept for many other applications, including mutivariate statistics (*used by all social scientists*), analysis of communications within a group (*see module 0045 "COMMUNICATION MATRICES"*), electrical network problems, and mathematical models in psychology and economics.

Let's start with a *SPECIAL CASE: Multiplying A Matrix by a Vector.*

Consider the following set of three linear equations in three unknowns:

$$\begin{aligned} 4x_1 + 7x_2 + 12x_3 &= 95 \\ 3x_1 + 2x_2 + 10x_3 &= 58 \\ 11x_1 + 8x_2 + 5x_3 &= 121 \end{aligned}$$

Matrix multiplication is defined in such a way that the above set of equations can be expressed in compact <u>matrix</u> form as follows:

$$A * X = B$$

Where: 1. "*" is the symbol for matrix multiplication

2. A is the matrix of coefficients

$$A = \begin{bmatrix} 4 & 7 & 12 \\ 3 & 2 & 10 \\ 11 & 8 & 5 \end{bmatrix}$$

3. B is the column vector of constant terms. *A column vector is a matrix with only one column.*

$$B = \begin{bmatrix} 95 \\ 58 \\ 121 \end{bmatrix}$$

4. X is the column vector of unknowns.

$$X = \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}$$

NOTE: Since X is a matrix its elements should have two subscripts--the first for the row and the second for the column--that is, strictly speaking we should write:

$$X = \begin{bmatrix} x_{11} \\ x_{21} \\ x_{31} \end{bmatrix}$$

But since there is only one column there should be no confusion if we drop the second (column) subscript.

That is, matrix multiplication is defined so that:

$$\begin{bmatrix} 4 & 7 & 12 \\ 3 & 2 & 10 \\ 11 & 8 & 5 \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} = \begin{bmatrix} 95 \\ 58 \\ 121 \end{bmatrix}$$

is a true statement.

But notice that

$$95 = 4x_1 + 7x_2 + 12x_3$$
$$58 = 3x_1 + 2x_2 + 10x_3$$
$$121 = 11x_1 + 8x_2 + 5x_3$$

hence, what we have said is that

$$\begin{bmatrix} 4 & 7 & 12 \\ 3 & 2 & 10 \\ 11 & 8 & 5 \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} = \begin{bmatrix} 4x_1 + 7x_2 + 12x_3 \\ 3x_1 + 2x_2 + 10x_3 \\ 11x_1 + 8x_2 + 5x_3 \end{bmatrix}$$

is a true statement by the definition of matrix multiplication.

In general, if

$$A = \begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix} \quad \textit{3 rows}$$

then

$$A*X = \begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} \quad \textit{3 rows}$$

1 column

$$= \begin{bmatrix} a_{11}x_1 + a_{12}x_2 + a_{13}x_3 \\ a_{21}x_1 + a_{22}x_2 + a_{23}x_3 \\ a_{31}x_1 + a_{32}x_2 + a_{33}x_3 \end{bmatrix} \Bigg\} \textit{3 rows}$$

1 column

NOTE: This has only <u>one</u> column.

Notice that the product of a *3 x 3 matrix and a 3 x 1 matrix (column vector) is a 3 x 1 matrix (column vector)* which is formed as follows:

Multiply each number in the *first row* of A by the corresponding number (*first by first, second by second, third by third*) in the *first column* of X (*in this case X only has one column*) and add these products. This gives the number in the first row and first column of A * X.

$$\text{row 1} \begin{bmatrix} a_{11} & a_{12} & a_{13} \\ \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}_{\text{column 1}} = \begin{bmatrix} a_{11}x_1 + a_{12}x_2 + a_{13}x_3 \\ \cdot \quad \cdot \quad \cdot \\ \cdot \quad \cdot \quad \cdot \end{bmatrix} \text{(row 1 column 1)}$$

Do the same for the *second row* of A and the *first, and only, column* of X to determine the number in the second row and first column of A *X. That is:

$$\text{row 2} \begin{bmatrix} \cdot & \cdot & \cdot \\ a_{21} & a_{22} & a_{23} \\ \cdot & \cdot & \cdot \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}_{\text{column 1}} = \begin{bmatrix} \cdot \\ a_{21}x_1 + a_{22}x_2 + a_{23}x_3 \\ \cdot \end{bmatrix} \text{(row 2 column 1)}$$

Finally the number in the *third row* and the *first column* of A * X is computed as follows:

$$\text{row 3} \begin{bmatrix} \cdot & \cdot & \cdot \\ \cdot & \cdot & \cdot \\ a_{31} & a_{32} & a_{33} \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}_{\text{column 1}} = \begin{bmatrix} \cdot \\ \cdot \\ a_{31}x_1 + a_{32}x_2 + a_{33}x_3 \end{bmatrix} \text{(row 3 column 1)}$$

To repeat the final result:

$$\overbrace{\begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix}}^{A} * \overset{X}{\begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}} = \overbrace{\begin{bmatrix} a_{11}x_1 + a_{12}x_2 + a_{13}x_3 \\ a_{21}x_1 + a_{22}x_2 + a_{23}x_3 \\ a_{31}x_1 + a_{32}x_2 + a_{33}x_3 \end{bmatrix}}^{A*X}$$

NOTE: To keep track of what you are doing, it is sometimes helpful to point to the elements that are being multiplied.

For example, when you are computing the number in row 2 column 1 of A * X above,

Your left hand points to:

$$\boxed{a_{21} \quad a_{22} \quad a_{23}}$$

Your right hand points to:

$$\begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}$$

and you think of placing them side by side and combining them as follows:

$$\begin{matrix} x_1 & & x_2 & & x_3 \\ * & + & * & + & * \\ a_{21} & & a_{22} & & a_{23} \end{matrix} \longrightarrow \begin{bmatrix} a_{21}x_1 + a_{22}x_2 + a_{23}x_3 \\ \\ \end{bmatrix}$$

etc.

Try it--you'll like it!

EXERCISE 1A | $A * X = B$ |

A program in BASIC which multiplies 3 x 3 matrix* by a 3 x 1 column vector might be organized as follows:



Write the complete program and try it with the following matrices:

$$N = 3, \quad \begin{bmatrix} 3 & 1 & 7 \\ 9 & 5 & 2 \\ 2 & 4 & 8 \end{bmatrix} * \begin{bmatrix} -1 \\ 6 \\ 4 \end{bmatrix} \qquad N = 3, \quad \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} * \begin{bmatrix} 20 \\ 20 \\ 20 \end{bmatrix}$$

Also multiply these by hand as well as by using your program.

---

* 3 x 3 is "pronounced" 3 by 3; the x does <u>not</u> mean multiplication.

EXERCISE 1B

In the above discussion there was nothing special about the number 3. The more general definition for multiplying a N x N matrix by a N x 1 matrix *(column vector)* is:

$$\begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ & \vdots & & \\ a_{n1} & a_{n2} & \cdots & a_{nn} \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{bmatrix} = \begin{bmatrix} a_{11}x_1 + a_{12}x_2 + \cdots + a_{1n}x_n \\ a_{21}x_1 + a_{22}x_2 + \cdots + a_{2n}x_n \\ \vdots \\ a_{n1}x_1 + a_{n2}x_2 + \cdots + a_{nn}x_n \end{bmatrix}$$

Write a BASIC program which reads a number N, a N x N matrix, and a N x 1 matrix, computes the matrix product, and prints the result.

Programming Note: A feature of BASIC which is useful in dealing with matrices is SUBCRIPTED VARIABLES. See your BASIC reference manual. You may also want to review FOR-NEXT loops. You will have to decide on a maximum possible value for N and write a DIM statement accordingly.

EXERCISE 1C

Test your program on the following data:

(a)

$$\begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 6 & 7 & 8 & 9 & 10 \\ 11 & 12 & 13 & 14 & 15 \\ 16 & 17 & 18 & 19 & 20 \\ 21 & 22 & 23 & 24 & 25 \end{bmatrix} * \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \end{bmatrix}$$

N = 5

(b)

$$\begin{bmatrix} 21 & 73 & 84 & 96 \\ 18 & 98 & 41 & 32 \\ 26 & 47 & 93 & 41 \\ 22 & 27 & 68 & 74 \end{bmatrix} * \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

N = 4

The General Case: More Than One Column

Suppose in the product A * X = B, A is a 3 x 3 matrix and X is a 3 x 2 matrix. Matrix multiplication is defined so that B will be a 3 x 2 matrix defined such that:

> Each column of the product matrix B is the column vector that you get by multiplying A by the corresponding column of X.

This is illustrated very well on the cover of this module and it would be worth your time to study it carefully.

For those who like symbols, subscript, etc.

3 rows
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix} * \begin{bmatrix} x_{11} & x_{12} \\ x_{21} & x_{22} \\ x_{31} & x_{32} \end{bmatrix} =$$
3 columns *2 columns*

3 rows
$$\begin{bmatrix} a_{11}x_{11} + a_{12}x_{21} + a_{13}x_{31} & a_{11}x_{12} + a_{12}x_{22} + a_{13}x_{32} \\ a_{21}x_{11} + a_{22}x_{21} + a_{23}x_{31} & a_{21}x_{12} + a_{22}x_{22} + a_{23}x_{32} \\ a_{31}x_{33} + a_{32}x_{21} + a_{33}x_{31} & a_{31}x_{12} + a_{32}x_{22} + a_{33}x_{32} \end{bmatrix}$$
2 columns

> Matrix multiplication is defined for matrices A and X whenever the number of columns of A is equal to the number of rows of X. Another way to say this is that any N x K matrix can be multiplied by any K x M matrix. The product matrix will be N x M matrix.

The rule to remember is: [N x K] * [K x M] = [N x M]

must be the same

Matrices which can be multiplied are called *conformable*. Notice that it may be possible to multiply A * X but not possible to multiply X * A. (Can you think of an example?)

EXERCISE 2A

Write a program which inputs three numbers N, K, M, a matrix A with N rows and K columns, and another matrix B with K rows and M columns. The program should output the matrix product A * B.

EXERCISE 2 B

Try your program on those products below which are conformable.

a) N = 3, K = 3, M = 2

$$\begin{bmatrix} 5 & 4 & 2 \\ 3 & 9 & 8 \\ 6 & 2 & 1 \end{bmatrix} * \begin{bmatrix} 2 & 7 \\ 5 & 3 \\ 6 & 4 \end{bmatrix} = ?$$

b) N = 3, K = 3, M = 3

$$\begin{bmatrix} 2 & 0 & 2 \\ -5 & 3 & 6 \\ 7 & 1 & 8 \end{bmatrix} * \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} = ?$$

*EXTRA: An element I (in a set with a multiplication operation defined on it) such that I * A = A * I = A for any A is called a multiplicative identity. What is the multiplicative identify for N x N (square) matrices? What is the additive identity?*

Answers for N = 2

MULTIPLICATIVE Identity $= \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$ ADDITIVE Identity $= \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$

c) N = 4, K = 4, M = 6

$$\begin{bmatrix} 1 & 5 & 1 & 5 \\ -1 & 3 & -1 & 3 \\ 2 & 0 & 5 & 6 \\ -7 & 1 & 3 & 8 \end{bmatrix} * \begin{bmatrix} -3 & 1 & 0 & 2 & 4 & -2 \\ 6 & 4 & -5 & 12 & 8 & 3 \\ 7 & 6 & -1 & -3 & 5 & 9 \\ 0 & 1 & 5 & 9 & 7 & \end{bmatrix} = ?$$

The above two matrices are not conformable for multiplication in the reverse oder. Hence, multiplication is not *commutative* for every pair of matrices. *(That is, it is not true that A * B = B * A for all matrices A and B.)* But what if A and B are N x N (square) matrices? Do you think that matrix multiplication is commutative for square matrices?

d) N = 2, K = 2, M = 2

$$\begin{bmatrix} 1 & 3 \\ -1 & 2 \end{bmatrix} * \begin{bmatrix} 1 & 0 \\ 2 & 1 \end{bmatrix} = ?$$

What can you conclude from (d) and (e)?

e) N = 2, K = 2, M = 2

$$\begin{bmatrix} 1 & 0 \\ 2 & 1 \end{bmatrix} * \begin{bmatrix} 1 & 3 \\ -1 & 2 \end{bmatrix} = ?$$

f) N = K = M = 3

$$\begin{bmatrix} 1.000000 & 0.500000 & 0.333333 \\ 0.500000 & 0.333333 & 0.250000 \\ 0.333333 & 0.250000 & 0.200000 \end{bmatrix} * \begin{bmatrix} 9 & -36 & 30 \\ -36 & 192 & -180 \\ 30 & -180 & 180 \end{bmatrix}$$

LISTING OF A GENERAL PROGRAM TO MULTIPLY MATRICES

```
10 PRINT "THIS IS A PROGRAM TO MULTIPLY 2 MATRICES."
20 PRINT "THE MATRICES MUST BE CONFORMABLE & OF"
30 PRINT "SIZE 10 BY 10 OR LESS."
40 PRINT "INPUT THE DIMENSIONS OF THE MATRIX TO BE MULTIPLIED."
50 PRINT "WHAT IS THE NUMBER OF ROWS?"
60 INPUT R1
70 PRINT "WHAT IS THE NUMBER OF COLUMNS?"
80 INPUT C1
90 PRINT "INPUT THE DIMENSIONS OF THE MULTIPLIER MATRIX."
100 PRINT "WHAT IS THE NUMBER OF ROWS?"
110 INPUT R2
120 IF R2=C1 THEN 150
130 PRINT "THE MATRICES ARE NOT CONFORMABLE."
140 GOTO 90
150 PRINT "WHAT IS THE NUMBER OF COLUMNS?"
152 INPUT C2
155 PRINT "INPUT THE MATRIX TO BE MULTIPLED."
160 PRINT "INPUT BY ROWS, ONE ELEMENT AT A TIME."
170 FOR I=1 TO R1
180 FOR J=1 TO C1
190 INPUT A(I,J)
200 NEXT J
210 NEXT I
220 PRINT "INPUT THE MATRIX THAT WILL BE THE MULTIPLIER."
230 PRINT "INPUT BY ROWS, ONE ELEMENT AT A TIME."
240 FOR I=1 TO R2
250 FOR J=1 TO C2
260 INPUT B(I,J)
270 NEXT J
289 NEXT I
300 FOR I=1 TO R1
310 FOR J=1 TO C2
320 LET C(I,J)=0
330 NEXT J
340 NEXT I
360 FOR K=1 TO C2
370 FOR I=1 TO R1
380 FOR J=1 TO R2
390 LET C(I,K)=C(I,K)+A(I,J)*B(J,K)
400 NEXT J
410 NEXT I
415 NEXT K
420 PRINT "THE PRODUCT MATRIX IS"
430 PRINT
440FOR I=1 TO R1
450 FOR J=1 TO C2
460 PRINT C(I,J);
470 NEXT J
480 PRINT
490 NEXT I
500 PRINT
510 PRINT "DO YOU WISH TO DO ANOTHER MULTIPLICATION?"
520 PRINT "IF SO TYPE A 1 OTHERWISE TYPE A 2."
530 INPUT Y
540 IF Y=1 THEN 40
550 END
```

SAMPLE RUN OF /MAT/ USING THE DATA FROM THE COVER PICTURE

```
>RUN
THIS IS A PROGRAM TO MULTIPLY 2 MATRICES.
THE MATRICES MUST BE COMPATIBLE AND OF
SIZE 10 BY 10 OR LESS.
INPUT THE DIMENSIONS OF THE MATRIX TO BE MULTIPLIED.
WHAT IS THE NUMBER OF ROWS?
? 3
WHAT IS THE NUMBER OF COLUMNS?
? 3
INPUT THE DIMENSIONS OF THE MULTIPLIER MATRIX.
WHAT IS THE NUMBER OF ROWS?
? 3
WHAT IS THE NUMBER OF COLUMNS?
? 2
INPUT THE MATRIX TO BE MULTIPLED.
INPUT BY ROWS, ONE ELEMENT AT A TIME.
? 5
? 4
? 2
? 3
? 9
? 8
? 6
? 2
? 1
INPUT THE MATRIX THAT WILL BE THE MULTIPLIER.
INPUT BY ROWS, ONE ELEMENT AT A TIME.
? 2
? 7
? 5
? 3
? 6
? 4
THE PRODUCT MATRIX IS

 42  55
 99  80
 28  52

DO YOU WISH TO DO ANOTHER MULTIPLICATION?
IF SO TYPE A 1 OTHERWISE TYPE A 2.
? 2

>EXIT
```

Communication Matrices

PROJECT SOLO
Department of Computer Science, University of Pittsburgh
Pittsburgh, Pennsylvania (15213) Module 0045

(FEATURING THE CASE OF THE MUTATED VIRUS ON PLANET X!)

● This module shows the application of matrices in studying communication patterns. Problems related to airline ticketing, message delivery, and communicable diseases will be presented.

● The module "Matrix Multiplication" is a prerequisite for doing the problems in this module.



The Airline Ticketing Problem:

EAGLE Airlines, a little known but ambitious "scheduled" commuter service, lists the following direct flights:

Eichelbergertown	to	Salladasburg
Eichelbergertown	to	Klingerstown
Eichelbergertown	to	Lackawaxen
Applebachsville	to	Dead Man's Gulch
Applebachsville	to	Klingerstown
Lackawaxen	to	Applebachsville
Lackawaxen	to	Klingerstown
Lackawaxen	to	Salladasburg
Satchelville	to	Klingerstown
Satchelville	to	Eichelbergertown
Salladasburg	to	Eichelbergertown
Klingerstown	to	Satchelville
Klingerstown	to	Salladasburg

An interesting question we can ask is the following:

> For any given starting city, how many different ways can you fly to each of the cities (including the starting city) via EAGLE Airlines?

(We are not concerned with how long a passenger may have to wait for a connection in this problem).

Diagrammatically we can represent EAGLE's flight schedule as shown below:



A diagram such as the above is called a directed graph. The direction of the arrows represents a flight from one town to another.

(NOTE: We're assuming that there are no flights scheduled that take off, circle overhead, then land in the very same town! Such a connection would be represented as:

Such connections are useful in some applications. After finishing this module, see if you can think of any such applications. Also check the modules on "Finite State Automata.")

Another way to represent the EAGLE flight routes is with a rectangular array:

TO: FROM	EICHELBERGERTOWN	APPLEBACHSVILLE	LACKAWAXEN	SATCHELVILLE	SALLADASBURG	KLINGERSTOWN	DEAD MAN'S GULCH
EICHELBERGERTOWN	NO	NO	YES	NO	YES	YES	NO
APPLEBACHSVILLE	NO	NO	NO	NO	NO	YES	YES
LACKAWAXEN			NO				
SATCHELVILLE				NO			
SALLADASBURG					NO		
KLINGERSTOWN	NO	NO	NO	YES	YES	NO	
DEAD MAN'S GULCH							

Either "YES" or "NO" is put in each box in answer to the question "Is there a direct flight from (row city) to (column city)?"

A part of the table has been filled in. Complete the table by looking at EAGLES's schedule on page 2, placing a YES or NO in each space.

When using the computer to help solve Problem 1, the table above is represented as a matrix. The rows and columns of the matrix correspond to the rows and columns of the table. The elements of the matrix are 0 to represent NO and 1 to represent YES.

The matrix representation would be:

$$A = \begin{bmatrix} 0 & 0 & 1 & 0 & 1 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 & 1 & 1 & 0 \\ 1 & 0 & 0 & 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

By representing the flight schedule as this binary matrix we can extract further information about the schedule. The matrix operations of addition and multiplication are used in the investigation.

In the binary matrix the non-zero entries represent a <u>direct</u> connection from one town to another. This direct connection will be called a <u>1-path</u>, or a path of <u>length-1</u>. By squaring this matrix (raising it to the 2nd power i.e. calculating A*A) we produce a matrix whose entries are the number of <u>2-paths</u>, or <u>paths of length 2</u>, from one town to another. Such a path of <u>length 2</u> requires 2 direct connection flights to get to the destination.

Squaring the matrix A gives:

$$\begin{array}{ccccccc} a & b & f & c & e & d & g \end{array}$$

$$A^2 = A*A = \begin{bmatrix} 1 & 1 & 0 & 1 & 2 & 1 & 0 \\ 0 & 0 & 0 & 1 & 1 & 0 & 0 \\ 1 & 0 & 0 & 1 & 1 & 1 & 1 \\ 0 & 0 & 1 & 1 & 2 & 1 & 0 \\ 0 & 0 & 1 & 0 & 1 & 1 & 0 \\ 2 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

(NOTE: A*A was calculated using the method explained in the module "MATRIX MULTIPLICATION".)

First row of A^2

WHAT DOES THIS NEW MATRIX TELL US?

Look at the first row. The numbers there represent the flight originating from Eichelbergertown. There is one 2-path to each of the towns: Eichelbergertown, Applebachsville, Satchelville and Klingerstown, as follows:

a. Eichelbergertown to Salladsburg to Eichelbergertown

b. Eichelbergertown to Lackawaxen to Applebachsville

c. Eichelbergertown to Klingerstown to Satchelville

d. Eichelbergertown to Lackawaxen to Klingerstown

There are two 2-paths to Salladsburg:

e. { Eichelbergertown to Lackawaxen to Salladsburg
Eichelbergertown to Klingerstown to Salladsburg }

There are no 2-paths to Lackawaxen (f) or to Dead Man's Gulch (g).

What do you think the entries of the matrix which results from cubing the original matrix represent?

We will use the mathematical notation A^2 to mean A*A, and A^3 to mean A*A*A. Note that A^3 also = A^2*A.

In our example with seven cities (in general n cities) it should be "obvious" that a passenger can get from one city to another city with six (in general n-1) or less direct flights.* If we also include round trips (that is, trips which return the passenger to his starting city), we should count the paths of length n or less. Thus in Problem 1 we're interested in counting all the paths of length n or less that connect any given starting city with any other city.

* Assuming he can get there at all.

One method of obtaining this information is to look at the diagram on page 3 and count all of the different ways. However, it is very easy to miss some paths. A more sophisticated method is to sum all of the power matrices (1st power, 2nd power, etc.) from the 1st power to the nth power, where n is the number of cities (and therefore the number of rows (or columns) in the binary matrix A). [If you don't remember how to add matrices, work through the module "ELEMENTARY MATRIX OPERATIONS".]

In the mathematical notation this is

$$S = A^1 + A^2 + A^3 + \ldots + A^{n-1} + A^n$$

where S is obtained by adding the matrices on the right. The entries in the summation matrix are the <u>total</u> number of paths of any length (1 to n) connecting any two towns.

PROBLEMS FOR COMPUTER SOLUTION

Problem 1

Write a computer program which finds the number of different ways in which a person can fly from a given starting city to each of the other cities (including the starting city) via EAGLE Airlines.

Sample Solution

On page 8 we will show a flow chart describing one solution of this problem; on page 9 a listing of a BASIC program that implements this flow chart is given. This program does *not* use the matrix functions available in most versions of BASIC in order to show you how a professional programmer manipulates matrices.

Flow Chart, Problem 1



Sample Program

```
5 DIM A(7,7), B(7,7), C(7,7), S(7,7)
10 PR. "INPUT THE CONNECTION MATRIX ROW BY ROW"
15 FOR I= 1 TO 7
20 FOR J= 1 TO 7
25 INPUT A(I,J)
30 NEXT J
35 NEXT I

40 FOR I= 1 TO 7
45 FOR J= 1 TO 7
50 S(I,J) = A(I,J)
55 B(I,J) = A(I,J)
60 NEXT J
65 NEXT I

70 LET N= 2
75 FOR I= 1 TO 7
80 FOR J= 1 TO 7
*85 C(I,J) = B(I,1)*A(1,J)+B(I,2)*A(2,J)+B(I,3)*A(3,J)+B(I,4)*A(4,J)
           +B(I,5)*A(5,J)+B(I,6)*A(6,J)+B(I,7)*A(7,J)
90 NEXT J
95 NEXT I

100 FOR I= 1 TO 7
105 FOR J= 1 TO 7
110 S(I,J) = S(I,J)+C(I,J)
115 NEXT J
120 NEXT I

125 N=N+1
130 IF N>7 THEN 165

135 FOR I= 1 TO 7
140 FOR J= 1 TO 7
145 B(I,J) = C(I,J)
150 NEXT J
155 NEXT I
160 GOTO 75

165 PR. "THE SUMMATION MATRIX IS:"
170 FOR I= 1 TO 7
175 FOR J= 1 TO 7
180 PRINT S(I,J);
185 NEXT J
190 PRINT
195 NEXT I

200 END
```

The line numbers used here correspond to those shown on the flow chart of page 8.

*Can you code this step in a more "elegant" manner?

Problem 2

(a) The schedule of a larger airline company is given below. Calculate the number of paths a passenger can choose in travelling from Philadelphia to Hawaii, if he wants to make stops in any three cities before reaching his destination (Hawaii).

(b) On his return, he wishes to go from Hawaii to Washington, again stopping in any three cities before reaching Washington. How many choices does the passenger now have?



Problem 3

Communication matrices can also be used in investigating directed communication of messages.

In the newly formed World Organization Opposed to Polluting the Sea there are eight member countries. In appointing representatives to WOOPS the member countries governing officers overlooked their appointee's language qualifications. Thus each appointee cannot speak to all of the other representatives directly. It is also possible that some representatives are acting under a "closed-mouth" policy. This policy allows a representative to listen to one speaking to him but he, in turn, will not speak or respond to this particular representative.

EXAMPLE:

The Spanish representative speaks to the German representative.

The German representative speaks to the Spanish and the American representatives.

The Italian representative speaks to the American and the French representatives.

The Chinese representative speaks to the Russian representative.

The American representative speaks to the German and the Italian representatives.

The Nigerian representative speaks to the Russian representative.

The Russian representative speaks to the Chinese, the Nigerian, and the French representatives.

The French representative speaks to the Russian representative.

(In this example the French representative is the only one operating under the "closed-mouth" policy.)

Finally, let's assume that a message becomes hopelessly garbled if it is translated more than five times.

Problem:

The German representative wishes to argue the validity of throwing empty beer bottles overboard from his country's freighters. Can his statement be routed so that all representatives of WOOPS receive his message? Which countries can and which cannot get messages through to all representatives?

Problem 4

A strange new plague has swept over the countryside of Planet X. It is due to Virus V1, which arrived via a meteorite that has landed from outer space. The inhabitants of Planet X who contacted the meteorite directly contract a disease called D1. To survive they must receive an antidote called A1.

NOW FOR SOME BAD NEWS:

If an inhabitant with disease D1 contacts another inhabitant of X, the virus is transferred in mutated form as V2. The only antidote for V2 is called A2. Similarly, further transfer of the virus produces additional mutations with new antidotes required as follows:



DISEASE	D1	D2	D3	D4	NONE
ANTIDOTE	A1	A2	A3	A4	

It is possible for an inhabitant to be exposed to several viruses through various contacts, in which case he must receive <u>each</u> appropriate antidote. The <u>amount</u> of antidote to be given is proportional to the number of paths by which the virus was communicated, with 1 gram needed for each communication.

<u>AND NOW FOR SOME GOOD NEWS</u>:

It turns out that virus V5 (and all higher numbered viruses) are harmless. Thus only four antidotes (A1, A2, A3, and A4) are needed to treat any inhabitant of Planet X.

<u>EXAMPLE</u>:

Let's prescribe appropriate antidote dosages for the three inhabitants shown below.



The arrows indicate communication between inhabitants.

A^1 =

	M	①	②	③
M	0	1	0	1
①	0	0	1	0
②	0	1	0	0
③	0	0	1	0

A^2 =

	M	①	②	③
M	0	0	2	0
①	0	1	0	0
②	0	0	1	0
③	0	1	0	0

A^3 =

	M	①	②	③
M	0	2	0	0
①	0	0	1	0
②	0	1	0	0
③	0	0	1	0

A^4 =

	M	①	②	③
M	0	0	2	0
①	0	1	0	0
②	0	0	1	0
③	0	1	0	0

NOTE: Only the first row, (M) should be examined in this problem, since viruses can only originate with the meteorite M.

SOLUTION:

	A1	A2	A3	A4
INHABITANT 1	1 GRAM	0 GRAMS	2 GRAMS	0 GRAMS
INHABITANT 2	0 GRAMS	2 GRAMS	0 GRAMS	2 GRAMS
INHABITANT 3	1 GRAM	0 GRAMS	0 GRAMS	0 GRAMS

(CAN YOU FIND ALL THESE DISEASE COMMUNICATING PATHS?)

NOW BACK TO ...

PROBLEM 4:

Prescribe drugs for the 17 inhabitants of Planet X shown on the cover of this module. If you don't have a computer you have our sympathy!

Some selected answers to the above problems:

1. You should have found a total of 18 paths from Eichelbergertown to Applebachsville.

2. (b) There are five ways to go from Hawaii to Washington with stops in three cities:

Hawaii to Chicago to New York to Philadelphia to Washington
Hawaii to Chicago to Detroit to Philadelphia to Washington
Hawaii to Chicago to Cleveland to Pittsburgh to Washington
Hawaii to Chicago to Cleveland to Philadelphia to Washington
Hawaii to Seattle to New York to Philadelphia to Washington

4. The dosage for Inhabitant 5 is:

1 gram of A1, 1 gram of A2, 2 grams of A3, 3 grams of A4.

digital

DIGITAL EQUIPMENT CORPORATION, Maynard, Massachusetts, Telephone: (617) 897-5111 • ARIZONA, Phoenix • CALIFORNIA, Anaheim, Los Angeles, Oakland, Palo Alto, San Diego and San Francisco • COLORADO, Denver • CONNECTICUT, Meriden • DISTRICT OF COLUMBIA, Washington (Riverdale, Md.) • FLORIDA, Orlando • GEORGIA, Atlanta • ILLINOIS, Chicago • INDIANA, Indianapolis • LOUISIANA, New Orleans • MASSACHUSETTS, Cambridge and Waltham • MICHIGAN, Ann Arbor • MINNESOTA, Minneapolis • MISSOURI, St. Louis • NEW JERSEY, Parsippany and Princeton • NEW MEXICO, Albuquerque • NEW YORK, Centereach (L.I.), New York City, (Englewood, N.J.), and Rochester • NORTH CAROLINA, Durham/Chapel Hill • OHIO, Cleveland and Dayton • OREGON, Portland • PENNSYLVANIA, Philadelphia and Pittsburgh • TENNESSEE, Knoxville • TEXAS, Dallas and Houston • UTAH, Salt Lake City • WASHINGTON, Seattle • WISCONSIN, Milwaukee • ARGENTINA, Buenos Aires • AUSTRALIA, Adelaide, Brisbane, Melbourne, Perth and Sydney • BELGIUM, Brussels • CANADA, Edmonton, Alberta; Vancouver, British Columbia; Carleton Place, Ottawa and Toronto, Ontario; and Montreal, Quebec • CHILE, Santiago • FRANCE, Grenoble and Paris • GERMANY, Cologne, Hannover, Frankfurt and Munich • ITALY, Milan • JAPAN, Tokyo • NETHERLANDS, The Hague • NORWAY, Oslo • PHILIPPINES, Manila • SPAIN, Barcelona and Madrid • SWEDEN, Stockholm • SWITZERLAND, Geneva and Zurich • UNITED KINGDOM, Birmingham, Edinburgh, London, Manchester and Reading • VENEZUELA, Caracas